Empirical Small-Signal Modeling of Switching Converters Using Pspice

Phuong Huynh
Philips Laboratories
Briarcliff, NY 10510 USA

Bo H. Cho
Department of Electrical Engineering
Seoul National University
Seoul 151-742 Korea

Abstract-- The objective of this paper is to present a new technique for generating small-signal transfer functions of switching converters. The technique requires generation of the input-output data from a switching converter using any simulation software packages, such as Pspice, and applies a simple algorithm on the recorded data to identify an AutoRegressive Moving Average (ARMA) model for the switching converter. A small-signal transfer function can be obtained from the ARMA model thereby.

I. Introduction

Small-signal modeling of switching regulators has always been considered as a key parameter for feedback control implementation. Generation of converter small-signal models prior to hardware implementation is primordial because it can save time and development cost. In this paper, an efficient method for generating small-signal models for switching converters is presented.

Several approaches, such as state-space averaging [4], discrete modeling, and harmonics-balancing technique [3], have been proposed in an attempt to generalize small-signal models for switching converters. However, the state-space averaging method is limited only to PWM converters, and the other two approaches, even though applicable also to resonant converters, are too complicated and difficult to apply. The proposed modeling technique adopts an input-output data analysis point of view. It presumably treats any converter as a black box and identifies the unknown small-signal model of the converter based on its input/output response. The approach is efficient, accurate up to half of the switching frequency, and suitable for identification of various types of converters. As illustrated in Fig. 1, the small-signal modeling procedure consists of three steps: (1) Any converter can be modeled using Pspice for data generation, (2) a simple algorithm is performed on the data to approximate the equivalent AutoRegressive Moving Average (ARMA) model, and (3) conversion to transfer function is performed.

The paper is divided into two main sections. The first section addresses the theoretical underlying of the modeling technique, which includes data generation and development of the identification algorithm. The subsequent section illustrates applicability of the modeling technique through small-signal transfer function identification of a buck converter, a series resonant converter, and a forward multi-resonant converter.



Fig. 1: Identification Process

II. Methodology

A. Data Generation

Any switching converter of unknown structure can be modeled using Pspice; a disturbance of small amplitude is injected around a nominal equilibrium point of the input; and both the input perturbation and the corresponding output responses are sampled for identification. Consequently, selection of the sampling frequency and the perturbation signal is primordial, since the sampled data must represent adequately the linear dynamic response of the converter around the steady-state operating point.

Selection of the input perturbation signal is important because in order to identify a converter accurately, the input-output data must contain all the different modes of the converter. To achieve this, the spectrum of the input signal must span the entire range of frequencies of interest. A pseudo-random binay signal (PRBS) inherents such a property [2]. It is a binary signal that swiches randomly to either A or $-A$ at every Δt and the random sequence repeats every period T, as shown in Fig. 2. Figure 3 illustrates the spectrum of a PRBS, which consists of a train of diract deltas following the envelop of the square of a $Sa(t)$ function ($\sin(t)/t$).

0-7803-2730-6/95 $4.00 © 1995 IEEE



Fig. 2: Pseudorandom Binary Signal (PRBS)



Fig. 3: Spectrum of PRBS

Since the information collected from the unknown system is in the form of measured input-output data, $\{u,y\}$, which are recorded at a specific sampling period, T_d, the spectra of the measured signals u and y consist of replication of the true spectra of the signals u and y at multiple of the sampling frequency, $1/T_d$, a phenomenon known as sampling effect. If the sampling frequency is selected to be smaller than half of the bandwidths of the signals u and y, the replicated spectra will overlap, and aliasing will occur, resulting in erroneous characterization of the unknown system. Therefore, selection of the perturbation frequency for the PRBSs is important. If the perturbation period, Δt, is the same as the sampling period, T_d, Fig. 4(a) illustrates the spectrum of the input signal, $\mathcal{P}_u(f)$, and indicates that there is still considerable power within the input signal above the Nyquist frequency ($1/2\Delta t$) and aliasing errors will occur. However, setting $\Delta t = 2T_d$ would move the replicated spectrum toward higher frequency, and the amount of overlap between the original spectrum and the replicated spectrum would be significantly reduced, as illustrated in Fig. 4(b). Thus, the illustration in Fig. 4(b) suggests that the perturbation period, Δt, should be at least two times larger than T_d to avoid the problem of aliasing when sampling the input data.

A method for generating a PRBS is to use a shift register circuit (Fig. 5) which can be modeled in Pspice as shown in Appendix C. For a shift register circuit of n registers, a maximum random binary sequence period of $2^n - 1$ can be achieved by performing an XOR operation between the n^{th} register and a specific m^{th} register. For a 7-shift-register circuit, the XOR operation must be performed between the 7^{th} register and the 3^{rd} register to achieve a maximum random binary sequence of 127 bits. On the other hand, for a 9-shift-register circuit, the XOR operation is performed between the 9^{th} and the 4^{th} registers and a random binary sequence of 511 bits is generated thereby. These maximum random binary sequences are adequate since only about 500 sampled data points are needed for the identification process.



Fig. 4: Aliasing



Fig. 5: Shift Register Circuit

The next issue concerns selection of the sampling frequency, f_d. For a switching converter, the outputs are usually an output voltage or an input current, and often, they can be periodic waveforms with a fundamental switching period, T_s. When dealing with such periodic signals, selection of the sampling frequency must be made carefully. These waveforms can be represented by a Fourier series expansion,

$$y(t) = K + \sum_{n=1}^{n=\infty} a_n \cos nw_s t + \sum_{n=1}^{n=\infty} b_n \sin nw_s t, \qquad (1)$$

where $y(t)$ represents the converter output, and $w_s = 2\pi f_s$ is the switching frequency of the converter, and K is a constant. When the input is selected as the input voltage or output current of the switching regulator and is perturbed with a signal of bandwidth B, the Fourier coefficients of the output $y(t)$, a_n, b_n, and K, becomes time-varying, and

$$y(t) = K(t) + \sum_{n=1}^{n=\infty} a_n(t) \cos nw_s t + \sum_{n=1}^{n=\infty} b_n(t) \sin nw_s t. \qquad (2)$$

According to the amplitude modulation theory (Appendix A), the spectrum of $y(t)$ becomes

$$Y(f) = K(f) + \frac{1}{2}\sum_{n=1}^{n=\infty} A_n(f - nf_s) + A_n(f + nf_s) + \frac{1}{2}\sum_{n=1}^{n=\infty} B_n(f - nf_s) + B_n(f + nf_s) \qquad , \qquad (3)$$

which contains the basedband spectrum, $K(f)$, and other centered at multiple of the switching frequency, as illustrated in Fig. 6. Therefore, if the bandwidth of the input perturbation signal, B, is larger than $f_s/2$, it will excite the output



Fig. 6: Spectrum of Output y(t)

signal to have a bandwidth larger than $f_s/2$ and the spectra will overlap, resulting in a phenomenon known as aliasing.

If the perturbed input is the duty cycle, the same phenomenon occurs. Indeed, when the input signal $v(t)$ is perturbed (Fig. 7), the duty cycle, $d(t)$ may be expressed in the form of a Fourier series as

$$d(t) = \frac{A}{T_s}[D + v(t)] + \sum_n \frac{2A}{n\pi}\sin\left[\frac{n\pi}{T_s}[D + v(t)]\right]\cos(nw_s t) \quad , \quad (4)$$

where A is a constant and D is the steady-state duty cycle. A quick inspection of (4) indicates that PWMing is characterized by both amplitude modulation and angle modulation. Under the assumption that $v(t)$ is a small perturbation, $|v(t)| << 1$, $\sin[Kv(t)] \approx Kv(t)$ and $\cos[Kv(t)] \approx 1$, where K is an arbitrary small constant. Therefore, (4) becomes

$$d(t) \approx \frac{A}{T_s}[D + v(t)] + \sum_n [K_1 + K_2 v(t)]\cos(n2\pi f_s t), \quad (5)$$

where $K_1 = \frac{2A}{n\pi}\sin\left(\frac{n\pi D}{T_s}\right)$ and $K_2 = \frac{2A}{T_s}\cos\left(\frac{n\pi D}{T_s}\right)$.

Consequently, the spectrum of $d(t)$ is given by

$$D(f) = \frac{A}{T_s}[D\delta(f) + V(f)] + \sum_n \frac{K_1}{2}[\delta(f - f_s) + \delta(f + f_s)] + \sum_n \frac{K_2}{2}[V(f - f_s) + V(f + f_s)]. \quad (6)$$

Equation (6) indicates that PWMing results in replicated spectra of $V(f)$ at multiples of the switching frequency, f_s; and therefore, $D(f)$ has an infinite bandwidth. It follows from (3) that the replicated spectra of $Y(f)$ overlap.

Fortunately, switching converters inherent lowpass properties, which means that the system bandwidth is usually smaller than the switching frequency, and therefore, aliasing is not too severe. For PWM converters, whose bandwidths are much smaller than the switching frequencies, it is expected that aliasing has negligible effect on the small-signal identification. However, resonant converters are characterized by resonant frequencies which are near the switching frequencies. It follows that aliasing may result in erroneous identification, especially in the high frequency range.



Fig. 7: Pulse Width Modulation

The above analysis suggests that the linear response beyond $f_s/2$ cannot be recovered by sampling the system faster than f_s since the aliasing problem is inherent in the converter. Consequently, a sampling frequency equal to or smaller than the switching frequency of the converter should be used to determine the small-signal model of the converter.

B. System Identification

The sampled data are used for the identification process in the subsequent stage. One of the approaches is based on the least-square fit algorithm [1]. Qualitatively, the identification algorithm approximates the order of the unknown converter and put the measured input/output data in an AutoRegressive Moving Average (ARMA) model form, such as the following n^{th}-order ARMA model

$$\begin{aligned} y(k) &= a_0 + a_1 y(k-1) + \ldots + a_{n-1} y(k-n+1) \\ &+ b_0 u(k) + b_1 u(k-1) + \ldots + b_n u(k-n) \\ &+ e(k) \end{aligned} \quad , \quad (7)$$

where $e(k)$ represents the error between the exact response and the approximated response of the ARMA model. It is desirable to estimate the parameters $[a_0, a_1, \ldots, a_{n-1}, b_0, b_1, \ldots, b_n]$ in such a way that the identified ARMA model will mimic as close as possible the linear response of the unknown system based on the measured sequence $\{u,y\}$. Substituting N sampled data points of the set $\{u,y\}$ in (7) yields

$$\mathbf{Y} = \mathbf{U}\boldsymbol{\theta} + \mathbf{E}, \quad (8)$$

where

$$\mathbf{Y} = \begin{bmatrix} y(n) \\ y(n+1) \\ \vdots \\ y(N) \end{bmatrix},$$

$$\mathbf{U} = \begin{bmatrix} 1 & y(n-1) & \cdots & y(1) & u(n) & \cdots & u(0) \\ 1 & y(n) & \cdots & y(2) & u(n+1) & \cdots & u(1) \\ \vdots & \vdots & \vdots & \vdots & \vdots & \vdots & \vdots \\ 1 & y(N-1) & \cdots & y(N-n+1) & U(N) & \cdots & U(N-n) \end{bmatrix},$$

$\boldsymbol{\theta} = [a_0, a_1, \ldots, a_{n-1}, b_0, b_1, \ldots, b_n]^T$, and $\mathbf{E}$ is the error vector. Let S denote the sum square error,

$$S = \mathbf{E}^T\mathbf{E} = (\mathbf{Y}^T - \boldsymbol{\theta}^T\mathbf{U}^T)(\mathbf{Y} - \mathbf{U}\boldsymbol{\theta}), \quad (9)$$



Fig. 8: Open-Loop Buck Converter



Fig. 9: Control-to-Output TF

it is desirable to estimate θ such that S is minimized. This can be accomplished by taking the derivative of S with respect to θ and setting it to zero; namely

$$\frac{\delta S}{\delta \theta} = -2U^T Y + 2U^T U\theta = 0\,. \tag{10}$$

Then, one obtains

$$\left[U^T U\right]\theta = U^T Y\,. \tag{11}$$

Equation (11) represents a set of algebraic equations from which θ can be solved using the Gaussian elimination technique with partial pivoting or better techniques that can be found in [1]. Thus, the identification process consists of applying (11), which can be implemented in MATLAB, to the sampled data obtained from the Pspice simulations. Estimation of the order of the unknown converter can be performed based on trying various ARMA models of different orders and selecting the one that yields the lowest sum-square error, S. Once the ARMA model is obtained, transform to the S-domain can be applied, and the converter small-signal transfer function is obtained thereby.

III. Result Verification

To validate the method, a buck converter, a series resonant converter (SRC), and a forward multiresonant converter (FMRC) are selected for the small-signal transfer function identification, as illustrated in Figs. 8, 10, and 12. PBRSs of small amplitudes (about 2% - 5% of the dc value) are injected into the converters; the sampling frequencies are set at the switching frequencies; and 500 input-output data points are collected for all three cases. The MATLAB program in Appendix B is used to identify the small-signal transfer functions. Since the system orders are unknown, various

TABLE I

Log (RMS Error)

System Order	Buck	SRC	FMRC
1	-0.97	-2.06	-1.34
2	***-2.54***	-2.33	-1.94
3	-2.57	***-2.96***	-2.12
4	-2.58	-3.00	-2.16
5		-3.02	-2.18
6			***-2.32***
7			-2.35
8			-2.36

system orders are selected, and the identification results are illustrated in Table 1. Observe that as the system order is increased, the identification result becomes more accurate for each converter (smaller RMS error). However, beyond a certain order, the RMS error starts to level down, indicating that increasing the system order further does not produce much more accuracy. Thus, according to Table 1, the identified ARMA model for the buck converter is of second order, the one for the SRC is of third order, and the one for the FMRC is of sixth order. Notice that the identification results are in good agreement with the measurement results, as illustrated in Figs. 9, 11, and 13. However, the disagreement in the phase of the FMRC case could result from the aliasing problem as predicted.

IV. Conclusions

A new empirical data modeling method for determining small-signal transfer functions of switching regulators is proposed. The method requires generation of the input-output data of a switching converter using a simulation software such as Pspice, and utilizes the least-square fit algorithm to approximate an ARMA model for the converter. The result verification shows that the method is accurate up to half of the switching frequency and is efficient, simple, and suitable for all classes of switching converters.

References

[1] J. P. Norton, " An Introduction to Identification," Academic Press, London, 1986.

[2] K. Godfrey, " Perturbation Signals for System Identification, " Prentice Hall, New York, 1993.

[3] E. Yang and F. C. Lee, " Extended Describing Function Method for Small-Signal Modeling of Switching Power Circuits, " VPEC Seminar, Blacksburg, VA 1994.

[4] R. D. Middlebrook and S. Cuk, " A General Unified Approach to Modeling Switching Converter Power Stages, "IEEE PESC 1976, pp. 18-34.



Fig. 10: Open-Loop Series Resonant Converter



Fig. 11: Control-to-Output TF



Fig. 12:Open-Loop Forward Multiresonant Converter



Fig. 13: Control-to-Output TF

Appendix A: Modulation Theory

In this appendix, basic modulation theory, which is needed for the empirical data modeling of switching regulators, is presented. By definition, modulation is the process of imparting the source information onto a bandpass signal with a carrier frequency f_c by the introduction of amplitude and/or phase perturbation.

A. Amplitude modulation

Given the baseband source signal, $s(t)$, the amplitude-modulated signal, $m(t)$, with the carrier frequency f_c is given by

$$m(t) = s(t)\cos(w_c t) \tag{12}$$

where $w_c = 2\pi f_c$. Consequently, if $S(f)$ and $M(f)$ denote the frequency domain representation of the signals $s(t)$ and $m(t)$, respectively, then $M(f)$ is given by

$$M(f) = \frac{1}{2}[S(f+f_c) + S(f-f_c)], \tag{13}$$

which means that the modulated signal $m(t)$ contains the shifted-spectra of the baseband signal $s(t)$ onto the carrier frequency $\pm f_c$, as illustrated in Fig. 14

B. Narrowband Angle Modulation

In angle modulation, the modulated signal, $m(t)$, is given by

$$m(t) = A\cos[w_c t + s(t)], \tag{14}$$

which is a nonlinear function of the modulation. However, when $s(t)$ is restricted to small perturbation, $m(t)$ can be approximated by a linear function of the modulation. Indeed, (14) can be expanded to



Fig. 14: Amplitude modulation

$$m(t) = A\cos(w_c t)\cos[s(t)] - A\sin(w_c t)\sin[s(t)]. \tag{15}$$

Since $|s(t)| << 1$, $\cos[s(t)] \approx 1$ and $\sin[s(t)] \approx s(t)$; and therefore

$$m(t) \approx A\cos(w_c t) - As(t)\sin(w_c t), \tag{16}$$

which, in the frequency domain, becomes

$$M(f) = \frac{A}{2}[\delta(f-f_c) + \delta(f+f_c)] + j\frac{A}{2}[S(f-f_c) - S(f+f_c)]. \tag{17}$$

The spectrum of $M(f)$ from (17) is similar to the one in Fig. 14, except for the presense of two additional delta functions centered at $\pm f_c$.

Appendix B: Matlab Program For Small-Signal Identification

```
load fmrc.dat      % Load data
U = fmrc(:,1);     % U = input
Y = fmrc(:,2);     % Y = output
Fs = 88496;        % Sampling frequency

%....... Compute ARMA model...........
order = 6;
```

```
n  = length(Y) - order;
UU = U(1:n);
YY = Y(1:n);
for i = 2:order+1;
   UU = [U(i:n+i-1) UU];
   YY = [Y(i:n+i-1) YY];
end;
Ye = YY(:,1);
Ue = [ones(n,1) UU YY(:,2:order+1)];
W = (Ue'*Ue) \ (Ue' * Ye);
%>>>> W = ARMA model coefficients <<<

%...... transfer function in Z-domain...
numd = W(2:order+2)';
dend = [1  -W(order+3:length(W))'];

%....... Z-domain -> S-domain .........
[num,den] = d2cm (numd,dend,1/Fs,'zoh');

%....... Plot tranfer function ........
w = logspace (1,6,300);
[mag,ph] = bode (num,den,w);
db = 20 * log10 (mag);
semilogx (w/2/pi,db);
title ('Transfer Function Gain [dB]');
figure
semilogx (w/2/pi,ph);
title ('Transfer Function Phase')
```

APPENDIX C: PSPICE CODES

```
****************************************
* SAMPLE & HOLD CIRCUIT                *
* node <sample> -> sample when 1       *
* node <hold>   -> hold   when 1       *
****************************************
.subckt SH1 in out s h PARAMS:cap=5e-7
E1    (1,0)        (in,0)    1
SW1   (1,2)        (s,0)     SWITCH
C1    (2,0)        {cap}     IC  = 0
E2    (3,0)        (2,0)     1
SW2   (3,4)        (h,0)     SWITCH
C2    (4,0)        {cap}     IC  = 0
Vcc   (vcc,0)                1
SW3   (vcc,out)    (4,0)     SWITCH
Rout  (out,0)      100
.ends


****************************************
* REGISTER SHIFT CIRCUIT: 7 Stages     *
*  Ts = Period                         *
*  td = time delay before perturbation *
*  Output = zero mean, amplitude {-1,1}*
****************************************
.subckt RSC7 out PARAMS: Ts=10u, td=10u
Vclk0 (s,0) PULSE  0   1 {td}       {Ts/10}
{Ts/10} {Ts/10}  {Ts}
Vclk1 (h,0) PULSE  0   1 {td+Ts/2} {Ts/10}
{Ts/10} {Ts/10}  {Ts}
Vctrl (c,0) PULSE  1   0 {td+Ts/2} {Ts/1e3}
{Ts/1e3} 1e6     2e6
Vini  (i,0) PULSE  0.5 1 {td}       {Ts/1e3}
{Ts/1e3} 1e6     2e6
XSH1 XOR O1 s h SH1 PARAMS: cap={Ts/10}
XSH2 O1  O2 s h SH1 PARAMS: cap={Ts/10}
XSH3 O2  O3 s h SH1 PARAMS: cap={Ts/10}
XSH4 O3  O4 s h SH1 PARAMS: cap={Ts/10}
XSH5 O4  O5 s h SH1 PARAMS: cap={Ts/10}
XSH6 O5  O6 s h SH1 PARAMS: cap={Ts/10}
XSH7 O6  O7 s h SH1 PARAMS: cap={Ts/10}
EXor (0,XOR) VALUE = { 2*(v(O3) - 0.5)*(v(O7)
- 0.5) - 0.5}  ; XOR circuit
RXor (0,XOR) 1
* Init 1st bit = 1 by tieing it to Vini
Sout (O1,i)  (c,0)  SWITCH
Eout (out,0) POLY(1) O7 0  -1  2
Rout (out,0) 1
.ends
.MODEL SWITCH VSWITCH (RON=.01,ROFF=1e6)

****************************************
* OPEN-LOOP BUCK CONVERTER             *
****************************************
Vg     (1,0)    30
SW     (1,2)    (d,0)     SW1
Diode  (0,2)              DIODE
L      (2,3)    80u       IC = 13.58
Rc     (3,4)    0.05
C      (4,0)    300u      IC = 14.50
Rload  (3,0)    1
*....PWM................................
XRSC 99 RSC9 PARAMS: Ts=40u, td=10u
Vd   (vc,0)    0.5
RVd  (vc,0)    1
Vr   (r,0) PULSE 0 1 0 19.99u 0.01u 0 20u
Rr   (r,0) 1
Ec   (d,0) VALUE={1/(1+exp(-
1e5*(.05*V(99)+V(vc)-V(r))))}
Rd   (d,0)  1
.MODEL SW1 VSWITCH (RON=.0001,ROFF=1e6)
.MODEL DIODE  D ()
.tran  20u  10e-3  0   0.5u     UIC
.end

****************************************
*FULL BRIDGE SERIES RESONANT CONVERTER *
****************************************
Vg   (1,0)   62.15
SW1  (1,2)   (d1,0)     SW
D1   (2,1)              DIODE
SW2  (1,3)   (0,d2)     SW
D2   (3,1)              DIODE
SW3  (2,0)   (0,d1)     SW
D3   (0,2)              DIODE
SW4  (3,0)   (d2,0)     SW
D4   (0,3)              DIODE
*...Switch control.................
XPRBS p RSC7 PARAMS: Ts=46.88u, td=0
Vc (dc,0) 0.05
Rv (dc,0) 1
Ec (99,0) POLY(2) (p,0) (dc,0) 0 0.03 1
Rc (99,0) 1
V1 (t,0)PULSE (-1 1 0 .01u .01u 11.70u
23.44u)
Rd1 (t,d1)      1
Cd1 (d1,0)      5n
Vr r,0 PULSE 0 1 0 11.71u .01u 0 11.72u
Rr r,0 1
E1 c,0 VALUE = {2*(1/(1+exp(-1e5*(v(99)-
v(r)))) - 0.5)}
R1 c,0 1
E2 0,s POLY(2) (c,0) (d1,0) 0 0 0 0 1
Rd2 (s,d2)  1
Cd2 (d2,0)  5n
L (3,5)   197u      ic = -1.3169485
C (5,6)   51n       ic = 1.7245076E2
Db1 (2,7)   DIODE
Db2 (6,7)   DIODE
Db3 (8,2)   DIODE
Db4 (8,6)   DIODE
Cf  (7,8)   32u      ic = 4.5954197E1
Rload  (7,8)   24.86
Ecf  (21,0)  (7,8)  1
REcf (21,0)         1

.model DIODE  D()
.model SW VSWITCH (Ron=1m, Roff=1MEG, Von=1,
Voff=-1)
```

```
.TRAN  11.72u   5.86m  0  0.05u  UIC
.end

****************************************
* Voltage-controlled-variable-frequency*
* Ramp generator                       *
* Ramp voltage amplitude [-5 V,5 V)    *
* Rise time = v(in)/1e-6               *
* Fall time = -100*v(in)/1e-6          *
** *************************************
.subckt RGEN  in  out
Gic (0,out)   VALUE = [v(in)/(1+exp(-
v(ctrl)))-100*v(in)/(1+exp(v(ctrl)))]
C   out  0    1u       ic = -5
Xs  out  ctrl STRIG
.ends

****************************************
* SCHMITT TRIGGER CIRCUIT:             *
* Output [-10.4 V, 10.4 V]             *
* If input voltage > 5, output = -10.4V*
* If input voltage < -5,output =  10.4V*
****************************************
.subckt  STRIG  1  3
Rin   (1,2)  1MEG
Ein   (7,0)  (2,1) 1000
Rc    (7,3)  1
C     (3,0)  100n         ic = 0
Dz1   (3,4)  ZENER
Dz2   (0,4)  ZENER
R1    (3,2)  10K
R2    (2,0)  10K
.MODEL ZENER D (BV = 8.47)
.ends

****************************************
* Forward Multi-Resonant Converter     *
****************************************
EVo        (21,0)    (8,6)    1
RVo        (21,0)             1
Vg   (1,0)   35.2
Rs   (1,2)   0.63
L    (2,3)   45u       ic = -3.8805144
Lm   (3,4)   139u      ic = 1.0163970
Rm   (3,4)   7.3K
Cs   (4,0)   48.2n     ic = 7.8406708E1
Diode  (0,4)           DIODE
Sd   (4,0)  (d,0)      SW1
* .. Duty cycle control ...............
En (40,0) POLY(1) (p,0) 1.1085  -0.043
Rn (40,0) 1
*..Ramp generator
Es (50,0) POLY(1) (p,0) .905   .01
Xr (50,r) RGEN
Ed (d,0)  VALUE = [1/(1+exp(-1e5*(v(40)-
v(r))))]
Rd (d,0)      1
XPRBS p RSC7 PARAMS: Ts = 22.6u, td = 0

*.......... Ideal Tranformer............
RXFORM (4 , 6)  1MEG
FXFORM (3 , 4)  VXFORM          0.5
VXFORM (5x, 5)  0
EXFORM (5x, 6)  (3 , 4)         0.5

Cd (5,6)  661n         ic = 9.5960226
D1 (5,7)  DIODE
D2 (6,7)  DIODE
Lf (7,8)  82u          ic = 3.3184993
Rc (8,9)  206m
Cf (9,6)  34u          ic = 8.3736696
Rload     (8,6)         2.8152

.MODEL SW1 VSWITCH (Ron=0.1m Roff=1Meg)
.MODEL DIODE  D ()
.tran  11.3u  5650u  0  0.02u  UIC
.end
```